
1. Symbolic algebra in Factor
A tour of math.symbolic and its companion vocabularies, as forty-odd worked problems. Every 

Factor block here is run by misc/symbolic-tutorial/check.py, which compares what Factor prints 

against the ! => lines, so the outputs in this document are the real ones.

Expressions are written in postfix inside symbolic[ ... ]: numbers are numbers, + - * / ^ neg 

sqrt sin cos tan exp log and the inverse, hyperbolic and gamma functions build expressions, pi, 

e and inf are constants, ?x is a pattern variable, and any other token is a variable. expr. prints an 

expression in ordinary notation, while . prints the underlying value.

2. Expressions and simplification

2.1. Like terms and powers

𝑥 + 𝑥, 𝑥 ⋅ 𝑥 ⋅ 𝑥, 𝑥2

𝑥

Sums and products are simplified as they are built.

symbolic[ x x + ] expr.

! => 2*x

symbolic[ x x * x * ] expr.

! => x^3

symbolic[ x 2 ^ x / ] expr.

! => x

symbolic[ x x - ] .

! => 0

2.2. Exact arithmetic

Rationals stay exact, and a number multiplying a sum is distributed.

1
2

+ 1
3

= 5
6
, 2(𝑥 + 1) = 2𝑥 + 2

symbolic[ 1 2 / 1 3 / + ] .

! => 5/6

symbolic[ 2 x 1 + * ] expr.

! => 2*x + 2

symbolic[ e e 1 - - ] .

! => 1

2.3. Expanding

(𝑥 + 1)2, (𝑥 + 𝑦)(𝑥 − 𝑦)

symbolic[ x 1 + 2 ^ ] expand expr.

! => x^2 + 2*x + 1

symbolic[ x y + x y - * ] expand expr.

! => x^2 - y^2

2.4. Printing and reading back

An expression prettyprints as a literal that reads back in, while expr. gives ordinary notation.



symbolic[ 3 x sin * x + ] .

! => symbolic[ x 3 x sin * + ]

symbolic[ 3 x sin * x + ] expr.

! => x + 3*sin(x)

symbolic[ 1 x / ] expr.

! => 1/x

symbolic[ 2 3 x * / ] expr.

! => 2/(3*x)

2.5. Substituting and evaluating

𝑥2 + 1 at 𝑥 = 2, sin(𝜋
4
)
√

2

symbolic[ x 2 ^ 1 + ] { { T{ sym f "x" } 2 } } subs .

! => 5

symbolic[ x x sin + ] { { T{ sym f "x" } T{ sym f "y" } } } subs expr.

! => y + sin(y)

symbolic[ pi 4 / sin 2 sqrt * ] evalf .

! => 1.0

2.6. Exact angles

sin(𝜋
6
) = 1

2
, cos(𝜋

6
) =

√
3

2
, tan(𝜋

3
) =

√
3

The inverse functions give the same angles back.

symbolic[ pi 6 / sin ] .

! => 1/2

symbolic[ pi 6 / cos ] expr.

! => sqrt(3)/2

symbolic[ pi 3 / tan ] expr.

! => sqrt(3)

symbolic[ 1 2 / asin ] expr.

! => pi/6

symbolic[ 3 sqrt atan ] expr.

! => pi/3

3. Differentiation

3.1. The product rule

d
d𝑥

(𝑥 sin 𝑥) = 𝑥 cos 𝑥 + sin 𝑥

symbolic[ x x sin * ] symbolic[ x ] differentiate expr.

! => x*cos(x) + sin(x)

3.2. The chain rule

d
d𝑥

𝑒𝑥2 = 2𝑥𝑒𝑥2



symbolic[ x 2 ^ exp ] symbolic[ x ] differentiate expr.

! => 2*x*exp(x^2)

symbolic[ 3 x * 1 + sin ] symbolic[ x ] differentiate expr.

! => 3*cos(3*x + 1)

3.3. Quotients and powers

d
d𝑥

𝑥
𝑥2 + 1

, d
d𝑥

𝑥𝑥

symbolic[ x x 2 ^ 1 + / ] symbolic[ x ] differentiate expand expr.

! => -2*x^2/(x^2 + 1)^2 + 1/(x^2 + 1)

symbolic[ x x ^ ] symbolic[ x ] differentiate expr.

! => x^x*(log(x) + 1)

3.4. Inverse and hyperbolic functions

d
d𝑥

arctan 𝑥 = 1
𝑥2 + 1

, d
d𝑥

tanh 𝑥 = 1
cosh2 𝑥

symbolic[ x atan ] symbolic[ x ] differentiate expr.

! => 1/(x^2 + 1)

symbolic[ x asin ] symbolic[ x ] differentiate expr.

! => 1/sqrt(-x^2 + 1)

symbolic[ x tanh ] symbolic[ x ] differentiate expr.

! => 1/cosh(x)^2

3.5. Gradients and Jacobians

For the polar map (𝑟, 𝜃) ↦ (𝑟 cos 𝜃, 𝑟 sin 𝜃) the Jacobian determinant is the 𝑟 of d𝑥 d𝑦 = 𝑟 d𝑟 d𝜃.

CONSTANT: polar { T{ sym f "r" } T{ sym f "theta" } }

symbolic[ r theta cos * ] symbolic[ r theta sin * ] 2array polar jacobian

[ [ expr. ] each ] each

! => cos(theta)

! => -r*sin(theta)

! => sin(theta)

! => r*cos(theta)

CONSTANT: polar { T{ sym f "r" } T{ sym f "theta" } }

:: det2 ( matrix -- expr )

    matrix first first2 :> ( a b )

    matrix second first2 :> ( c d )

    a d s* b c s* s- ;

symbolic[ r theta cos * ] symbolic[ r theta sin * ] 2array polar jacobian

det2 pythagorean-rules rewrite expr.

! => r

3.6. Classifying a critical point

𝑓(𝑥, 𝑦) = 𝑥3 − 3𝑥𝑦 + 𝑦3



The gradient vanishes at (1, 1) and at the origin. The Hessian determinant settles which is which.

CONSTANT: vars { T{ sym f "x" } T{ sym f "y" } }

CONSTANT: f symbolic[ x 3 ^ 3 x * y * - y 3 ^ + ]

f vars gradient [ expr. ] each

! => 3*x^2 - 3*y

! => 3*y^2 - 3*x

f vars hessian [ [ expr. ] each ] each

! => 6*x

! => -3

! => -3

! => 6*y

With 𝑓𝑥𝑥 = 6 > 0 and determinant 27 > 0 at (1, 1) it is a local minimum; at the origin the 

determinant is negative, so that is a saddle.

CONSTANT: vars { T{ sym f "x" } T{ sym f "y" } }

CONSTANT: f symbolic[ x 3 ^ 3 x * y * - y 3 ^ + ]

:: det2 ( matrix -- expr )

    matrix first first2 :> ( a b )

    matrix second first2 :> ( c d )

    a d s* b c s* s- ;

f vars hessian det2 expr.

! => 36*x*y - 9

f vars hessian det2 { { T{ sym f "x" } 1 } { T{ sym f "y" } 1 } } subs .

! => 27

f vars hessian det2 { { T{ sym f "x" } 0 } { T{ sym f "y" } 0 } } subs .

! => -9

3.7. Derivatives left unevaluated

D writes a derivative down without computing it; doit computes it.

symbolic[ x 3 ^ x D ] expr.

! => D(x^3, x)

symbolic[ x 3 ^ x D ] doit expr.

! => 3*x^2

4. Antiderivatives

4.1. Powers, trigonometry and exponentials

∫ 𝑥2 d𝑥 = 𝑥3

3
, ∫ 1

𝑥
d𝑥 = log 𝑥

symbolic[ x 2 ^ ] symbolic[ x ] integrate expr.

! => x^3/3

symbolic[ 1 x / ] symbolic[ x ] integrate expr.

! => log(x)

symbolic[ x exp ] symbolic[ x ] integrate expr.

! => exp(x)



4.2. Linear arguments

∫ sin 2𝑥 d𝑥 = −cos(2𝑥)
2

, ∫ 1
2𝑥 + 1

d𝑥

symbolic[ 2 x * sin ] symbolic[ x ] integrate expr.

! => -cos(2*x)/2

symbolic[ 1 2 x * 1 + / ] symbolic[ x ] integrate expr.

! => log(2*x + 1)/2

symbolic[ 3 x * 1 + exp ] symbolic[ x ] integrate expr.

! => exp(3*x + 1)/3

4.3. An arc tangent and a logarithm

∫ 1
𝑥2 + 1

d𝑥 = arctan 𝑥, ∫ tan 𝑥 d𝑥

symbolic[ x 2 ^ 1 + -1 ^ ] symbolic[ x ] integrate expr.

! => atan(x)

symbolic[ x tan ] symbolic[ x ] integrate expr.

! => -log(cos(x))

4.4. Squares of sine and cosine

∫ sin2 𝑥 d𝑥 = 𝑥
2

− sin 2𝑥
4

symbolic[ x sin 2 ^ ] symbolic[ x ] integrate expr.

! => x/2 - sin(2*x)/4

symbolic[ x sin x cos * ] symbolic[ x ] integrate expr.

! => -cos(x)^2/2

4.5. When the derivative divides

∫ 2𝑥𝑒𝑥2 d𝑥 = 𝑒𝑥2

This one is found automatically: the rest of the product is a constant times the derivative of the 

inner expression.

symbolic[ 2 x * x 2 ^ exp * ] symbolic[ x ] integrate expr.

! => exp(x^2)

symbolic[ x x 2 ^ 1 + / ] symbolic[ x ] integrate expr.

! => log(x^2 + 1)/2

4.6. No elementary antiderivative

The integral comes back unevaluated rather than wrong.

symbolic[ x 2 ^ neg exp ] symbolic[ x ] integrate expr.

! => integrate(exp(-x^2), x)

symbolic[ x sin x / ] symbolic[ x ] integrate expr.

! => integrate(sin(x)/x, x)



5. Integration by hand

5.1. By parts, choosing u

∫ 𝑥𝑒𝑥 d𝑥 = 𝑥𝑒𝑥 − ∫ 𝑒𝑥 d𝑥

by-parts-u leaves the remaining integral unevaluated, so the step is visible; doit finishes it.

symbolic[ x x exp * ] symbolic[ x ] symbolic[ x ] by-parts-u expr.

! => x*exp(x) - integrate(exp(x), x)

symbolic[ x x exp * ] symbolic[ x ] symbolic[ x ] by-parts-u doit expr.

! => x*exp(x) - exp(x)

5.2. By parts, choosing dv

∫ 𝑥 log 𝑥 d𝑥, d𝑣 = 𝑥 d𝑥

Choosing d𝑣 instead fixes 𝑢 = log 𝑥, whose derivative is simple.

symbolic[ x x log * ] symbolic[ x ] symbolic[ x ] by-parts-dv expr.

! => x^2*log(x)/2 - integrate(x/2, x)

symbolic[ x x log * ] symbolic[ x ] symbolic[ x ] by-parts-dv doit expr.

! => x^2*log(x)/2 - x^2/4

5.3. The dv = 1 trick

∫ log 𝑥 d𝑥 = 𝑥 log 𝑥 − ∫ 1 d𝑥

For a single logarithm there is nothing else to be d𝑣, so take d𝑣 = d𝑥.

symbolic[ x log ] symbolic[ x ] 1 by-parts-dv expr.

! => x*log(x) - integrate(1, x)

symbolic[ x log ] symbolic[ x ] 1 by-parts-dv doit expr.

! => -x + x*log(x)

5.4. Repeated parts

∫ 𝑥2 sin 𝑥 d𝑥

Each application lowers the power of 𝑥 by one.

symbolic[ x 2 ^ x sin * ] symbolic[ x ] symbolic[ x 2 ^ ] by-parts-u expr.

! => -x^2*cos(x) - integrate(-2*x*cos(x), x)

symbolic[ x 2 ^ x sin * ] symbolic[ x ] integrate expr.

! => -x^2*cos(x) + 2*x*sin(x) + 2*cos(x)

5.5. A substitution you choose

∫ 2𝑥𝑒𝑥2 d𝑥, 𝑢 = 𝑥2



symbolic[ 2 x * x 2 ^ exp * ] symbolic[ x ] symbolic[ x 2 ^ ]

by-substitution expr.

! => exp(x^2)

symbolic[ x sin x cos * ] symbolic[ x ] symbolic[ x sin ]

by-substitution expr.

! => sin(x)^2/2

The integrand has to mention the substituted expression after dividing by 𝑢′: sin 𝑥/ cos 𝑥 fits 𝑢 =
cos 𝑥, while the equal expression tan 𝑥 does not.

symbolic[ x sin x cos / ] symbolic[ x ] symbolic[ x cos ]

by-substitution expr.

! => -log(cos(x))

symbolic[ x tan ] symbolic[ x ] symbolic[ x cos ] by-substitution expr.

! => integrate(tan(x), x)

6. Definite, improper and special integrals

6.1. Definite integrals

∫
𝜋

0
sin 𝑥 d𝑥 = 2, ∫

1

0
𝑥2 d𝑥 = 1

3

symbolic[ x sin ] symbolic[ x ] 0 symbolic[ pi ] definite-integrate .

! => 2

symbolic[ x 2 ^ ] symbolic[ x ] 0 1 definite-integrate .

! => 1/3

symbolic[ x exp ] symbolic[ x ] 0 1 definite-integrate expr.

! => e - 1

6.2. Substituting the limits

A definite substitution maps the limits to 𝑢(𝑎) and 𝑢(𝑏), so there is nothing to substitute back.

∫
1

0
2𝑥𝑒𝑥2 d𝑥 = 𝑒 − 1

symbolic[ 2 x * x 2 ^ exp * ] symbolic[ x ] 0 1 symbolic[ x 2 ^ ]

definite-by-substitution expr.

! => e - 1

symbolic[ x sin x cos * ] symbolic[ x ] 0 symbolic[ pi 2 / ]

symbolic[ x sin ] definite-by-substitution .

! => 1/2

6.3. To infinity

∫
∞

0
𝑒−𝑥 d𝑥 = 1, ∫

∞

1

1
𝑥2 d𝑥 = 1

symbolic[ x neg exp ] symbolic[ x ] 0 symbolic[ inf ] definite-integrate .

! => 1

symbolic[ x -2 ^ ] symbolic[ x ] 1 symbolic[ inf ] definite-integrate .



! => 1

symbolic[ x 2 ^ 1 + -1 ^ ] symbolic[ x ] 0 symbolic[ inf ]

definite-integrate expr.

! => pi/2

6.4. Divergence, and a singular endpoint

∫
∞

1

1
𝑥

d𝑥 = ∞, ∫
1

0
log 𝑥 d𝑥 = −1

The second integrand is unbounded at 0, and the endpoint is evaluated as a limit.

symbolic[ 1 x / ] symbolic[ x ] 1 symbolic[ inf ] definite-integrate expr.

! => inf

symbolic[ x log ] symbolic[ x ] 0 1 definite-integrate .

! => -1

6.5. The Gaussian integral

∫
∞

−∞
𝑒−(𝑥−5)2/9 d𝑥 = 3

√
𝜋

There is no elementary antiderivative, so this is a rule of its own.

symbolic[ x 2 ^ neg exp ] symbolic[ x ] symbolic[ inf neg ] symbolic[ inf ]

definite-integrate expr.

! => sqrt(pi)

symbolic[ x 5 - 2 ^ 9 / neg exp ] symbolic[ x ]

symbolic[ inf neg ] symbolic[ inf ] definite-integrate expr.

! => 3*sqrt(pi)

6.6. The gamma function

∫
∞

0
𝑥𝑛𝑒−𝑎𝑥 d𝑥 = Γ(𝑛 + 1)

𝑎𝑛+1

symbolic[ x 2 ^ x neg exp * ] symbolic[ x ] 0 symbolic[ inf ]

definite-integrate .

! => 2

symbolic[ x 2 x * neg exp * ] symbolic[ x ] 0 symbolic[ inf ]

definite-integrate .

! => 1/4

symbolic[ 5 gamma ] .

! => 24

symbolic[ 3 2 / gamma ] expr.

! => sqrt(pi)/2

6.7. Symmetry

∫
𝜋/4

0
log(1 + tan 𝑥) d𝑥 = 𝜋 log 2

8

Here 𝑓(𝑥) + 𝑓(𝜋/4 − 𝑥) is constant, so the integral follows from that constant and the width of the 

interval, with no antiderivative.



symbolic[ 1 x tan + log ] symbolic[ x ] 0 symbolic[ pi 4 / ]

definite-by-symmetry expr.

! => log(2)*pi/8

6.8. Differentiating under the integral sign

𝐹(𝑡) = ∫
1

0
(𝑡𝑥2 + 𝑥) d𝑥, 𝐹 ′(𝑡) = ∫

1

0
𝑥2 d𝑥

symbolic[ t x 2 ^ * x + ] symbolic[ x ] 0 1 symbolic[ t ]

feynman-derivative .

! => 1/3

symbolic[ t x 2 ^ * x + ] symbolic[ x ] 0 1 symbolic[ t ] 0

feynman-solve expr.

! => t/3 + 1/2

7. Limits and series

7.1. Indeterminate quotients

lim
𝑥→0

sin 𝑥
𝑥

= 1, lim
𝑥→0

1 − cos 𝑥
𝑥2 = 1

2

symbolic[ x sin x / ] symbolic[ x ] 0 limit .

! => 1

symbolic[ 1 x cos - x 2 ^ / ] symbolic[ x ] 0 limit .

! => 1/2

7.2. Zero times infinity

lim
𝑥→0

𝑥 log 𝑥 = 0

symbolic[ x x log * ] symbolic[ x ] 0 limit .

! => 0

7.3. Limits at infinity

lim
𝑥→∞

2𝑥2 + 3
𝑥2 − 1

= 2, lim
𝑥→∞

𝑥
𝑒𝑥 = 0

symbolic[ 2 x 2 ^ * 3 + x 2 ^ 1 - / ] symbolic[ x ] symbolic[ inf ] limit .

! => 2

symbolic[ x x exp / ] symbolic[ x ] symbolic[ inf ] limit .

! => 0

symbolic[ x atan ] symbolic[ x ] symbolic[ inf ] limit expr.

! => pi/2

7.4. One to the infinity

lim
𝑥→0

(1 + 𝑥)1/𝑥 = 𝑒, lim
𝑥→0

(1 + sin 5𝑥)7/𝑥 = 𝑒35

Writing 𝑓𝑔 as exp(𝑔 log 𝑓) turns this into a product the engine can already do.



symbolic[ 1 x + 1 x / ^ ] symbolic[ x ] 0 limit expr.

! => e

symbolic[ 1 5 x * sin + 7 x / ^ ] symbolic[ x ] 0 limit expr.

! => exp(35)

An exponent that tends to 0 is not indeterminate, and needs no such treatment.

symbolic[ 1 5 x * sin + x 7 / ^ ] symbolic[ x ] 0 limit .

! => 1

7.5. No limit

lim
𝑥→∞

sin 𝑥 does not exist

symbolic[ x sin ] symbolic[ x ] symbolic[ inf ] limit expr.

! => limit(sin(x), x, inf)

7.6. Maclaurin series

𝑒𝑥 ≈ 1 + 𝑥 + 𝑥2

2
+ 𝑥3

6

symbolic[ x exp ] symbolic[ x ] 3 maclaurin expr.

! => x^3/6 + x^2/2 + x + 1

symbolic[ x sin ] symbolic[ x ] 5 maclaurin expr.

! => x^5/120 - x^3/6 + x

symbolic[ 1 x + log ] symbolic[ x ] 3 maclaurin expr.

! => x^3/3 - x^2/2 + x

7.7. Expanding about another point

𝑒𝑥 about 𝑥 = 1

symbolic[ x exp ] symbolic[ x ] 1 2 taylor expr.

! => e + e*(x - 1) + e*(x - 1)^2/2

symbolic[ x 3 ^ 2 x 2 ^ * - 5 + ] symbolic[ x ] 4 maclaurin expr.

! => x^3 - 2*x^2 + 5

8. Rules, compilation and numerics

8.1. The Pythagorean identity

sin2 𝑥 + cos2 𝑥 = 1

A rule set rewrites matching subexpressions, in any order of terms, and leaves the rest alone.

symbolic[ x sin 2 ^ x cos 2 ^ + ] pythagorean-rules rewrite .

! => 1

symbolic[ y 1 + sin 2 ^ y 1 + cos 2 ^ + z + ] pythagorean-rules rewrite expr.

! => z + 1

symbolic[ x sin 2 ^ y cos 2 ^ + ] pythagorean-rules rewrite expr.

! => cos(y)^2 + sin(x)^2



8.2. Rules you write

A rule literal uses pattern variables written ?x.

symbolic[ a b + c * ] rule[ ?x ?y + ?z * => ?x ?z * ?y ?z * + ] 1array

rewrite expr.

! => a*c + b*c

symbolic[ x y 3 ^ * log ] log-expand-rules rewrite expr.

! => 3*log(y) + log(x)

symbolic[ x exp y exp * z * ] exp-rules rewrite expr.

! => z*exp(x + y)

8.3. Conditional rules

A rule can carry a test, here that the exponent is an integer.

symbolic[ x 2 ^ log x y ^ log + ]

symbolic[ ?a ?n ^ log ] symbolic[ ?n ?a log * ] [ "n" binding integer? ]

<conditional-rule> 1array rewrite expr.

! => 2*log(x) + log(x^y)

8.4. Compiling an expression

An expression becomes a quotation, or a compiled word, taking the variables’ values.

3 4 symbolic[ x 2 ^ y 2 ^ + ] { T{ sym f "x" } T{ sym f "y" } }

expr>quot call( x y -- r ) .

! => 25

2 symbolic[ x -3 ^ ] { T{ sym f "x" } } expr>quot call( x -- r ) .

! => 1/8

8.5. Numeric integration

∫
1

0
𝑒−𝑥2 d𝑥 ≈ 0.7468

symbolic[ x 2 ^ neg exp ] T{ sym f "x" } 0 1 nintegrate .

! => 0.7468241328202144

8.6. Checking an antiderivative numerically

Differentiating the result and evaluating both at a point is a quick check that a rule did the right 

thing.

symbolic[ x 2 ^ x sin * ] symbolic[ x ] integrate

symbolic[ x ] differentiate { { T{ sym f "x" } 0.7 } } subs evalf .

! => 0.31566666674646854

symbolic[ 0.7 2 ^ 0.7 sin * ] evalf .

! => 0.31566666674646854

